# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Daniel Wallace <danielwallace at gtmanfred dot com>

pkgname=python-cmd2
pkgver=2.4.3
_commit=031832a76b7a9e25d708153085d18d5366ff318d
pkgrel=2
pkgdesc="A tool for building interactive command line apps"
arch=('any')
url="https://github.com/python-cmd2/cmd2"
license=('MIT')
depends=('python-attrs' 'python-colorama' 'python-pyperclip' 'python-wcwidth')
makedepends=('git' 'python-setuptools-scm')
checkdepends=('python-pytest-cov' 'python-pytest-mock' 'python-pytest-runner' 'python-pytest-xdist'
              'vi')
source=("git+https://github.com/python-cmd2/cmd2.git#commit=$_commit")
sha512sums=('SKIP')

build() {
  cd cmd2
  python setup.py build
}

check() {
  cd cmd2
  python setup.py pytest
}

package() {
  cd cmd2
  python setup.py install --root="$pkgdir" --optimize=1
}

# vim:set ts=2 sw=2 et:
